A brief introduction to Mathematica

Karl W Broman

Department of Biostatistics
Johns Hopkins University

`www.biostat.jhsph.edu/~kbroman`

Why Mathematica?

C	computational efficiency
Perl	text/data manipulation
R	interactive analyses
Mathematica	symbolic algebra/calculus

Everything you could do back when you had just taken calculus, only accurately.

You might instead use Maple, but I have no experience with it.

Open source alternatives: Axiom, Maxima.

Preliminaries

- Command-line version: type `math`
  (I use this, and copy-and-paste from a text file.)
- GUI (with mathematica "notebooks"): type `mathematica`
- To exit: type `Quit`

First stuff

```
In[1]:= 5^12
Out[1]= 244140625

In[2]:= %1 ^ (1/12)
Out[2]= 5

In[3]:= % + a
Out[3]= 5 + a

In[4]:= L = 3
Out[4]= 3

In[5]:= L
Out[5]= 3
```

```
In[6]:= L = 3;


In[7]:= 20 L
Out[7]= 60

In[8]:= 2 m + 3 m
Out[8]= 5 m

In[9]:= %%
Out[9]= 60

In[10]:= %8
Out[10]= 5 m
```

Help

```
?Factor*
?FactorInteger
??FactorInteger

?*Plot*

?@   (* defined objects *)
```

Buy a book, such as Abell & Braselton, *Mathematica by Example*, 3rd ed.

Use Google.

Packages

Sometimes, you need to load a separate package. I don't recall ever needing this.

```
In[1]:= GramSchmidt[{{1,1,0}, {0,2,1}, {1,0,3}}]
Out[1]= GramSchmidt[{{1, 1, 0}, {0, 2, 1}, {1, 0, 3}}]

In[2]:= Remove[GramSchmidt]

In[3]:= << LinearAlgebra`Orthogonalization`

In[4]:= GramSchmidt[{{1,1,0}, {0,2,1}, {1,0,3}}]

                1        1                 1         1        1
Out[4]= {{-------, -------, 0}, {-(-------), -------, -------},
          Sqrt[2]  Sqrt[2]         Sqrt[3]   Sqrt[3]  Sqrt[3]

          1           1           2
>    {-------, -(-------), Sqrt[-]}}
      Sqrt[6]     Sqrt[6]         3
```

A bit of notation

`[ ]`	Arguments to functions
`{ }`	Lists
`[[ ]]`	Subsetting lists

Numbers

```
In[1]:= 5 * 10
Out[1]= 50

In[2]:= 5 10
Out[2]= 50

In[3]:= a10
Out[3]= a10

In[4]:= a 10
Out[4]= 10 a
```

```
In[5]:= 1/2 + 2/144

        37
Out[5]= --
        72

In[6]:= 1/2 + 2.0/144
Out[6]= 0.513889
```

```
In[7]:= Sqrt[27]
Out[7]= 3 Sqrt[3]

In[8]:= Sqrt[27.0]
Out[8]= 5.19615

In[9]:= N[Sqrt[27]]
Out[9]= 5.19615
```

Constants

```
In[1]:= E
Out[1]= E

In[2]:= Pi
Out[2]= Pi

In[3]:= N[E, 25]
Out[3]= 2.718281828459045235360287

In[4]:= N[Pi, 100]
Out[4]= 3.1415926535897932384626433832795028841971693993751058209749445923078\
>     16406286208998628034825342117068
```


Algebra

Expand, Factor, Together, Apart, Simplify, FullSimplify

```
In[1]:= Expand[ (x + 2y + z)^2 ]
         2              2                    2
Out[1]= x  + 4 x y + 4 y  + 2 x z + 4 y z + z
```

```
In[2]:= Factor[ % ]
                     2
Out[2]= (x + 2 y + z)
```

```
In[3]:= Together[ 1/(1+2x) - 2/(2+3x) ]
                          x
Out[3]= -(-------------------)
          (1 + 2 x) (2 + 3 x)
```

```
In[4]:= Apart[ % ]
            1         2
Out[4]= ------- - -------
        1 + 2 x   2 + 3 x
```

Solving equations

`Solve`, `NRoots`

```
In[1]:=  f = x^3 - 3 x^2 - 17x + 51;


In[2]:= soln = Solve[f == 0, x]

Out[2]= {{x -> 3}, {x -> -Sqrt[17]}, {x -> Sqrt[17]}}


In[3]:= f /. soln

Out[3]= {0, 0, 0}


In[4]:= NRoots[ f == 0, x ]

Out[4]= x == -4.12311 || x == 3. || x == 4.12311
```


A silly example

Take $V_g = a^2/2 + d^2/4$, $V_e = V_g(1 - h^2)/h^2$, and $a = 4d$.

Supposing $V_e = 1$ and $h^2 = 0.6$, solve for d.

```
In[1]:= Vg = a^2 / 2 + d^2 / 4;
In[2]:= Ve = Vg (1-hsq)/hsq;
In[3]:= a = 4d;
In[4]:= hsq = 6/10;


In[5]:= Solve[Ve == 1, d]
                    2                2
Out[5]= {{d -> -Sqrt[--]}, {d -> Sqrt[--]}}
                    11               11

In[6]:= N[ % ]
Out[6]= {{d -> -0.426401}, {d -> 0.426401}}
```

Solving systems

Suppose we have $2p_1 + 2p_2 = 1$ and $p_1 = (1-r)p_1 + p_2/2$.

Solve for p_1 and p_2.

```
In[1]:= eqn1 = 2 p1 + 2 p2 == 1;
In[2]:= eqn2 = p1 == (1-r) p1 + p2 / 2;


In[3]:= Solve[ {eqn1, eqn2}, {p1, p2}]

                     1                r
Out[3]= {{p1 -> -----------, p2 -> -------}}
                2 (1 + 2 r)        1 + 2 r
```


A nonlinear example

Suppose $x^2 = 2y + 2$ and $x = y^2 + 1$.

Solve for x and y.

```
In[1]:= N [ Solve[ {x^2 == 2y + 2, x == y^2 + 1}, {x,y} ] ]


Out[1]= {{x -> 2., y -> 1.}, {x -> 1.1304, y -> -0.361103},

>     {x -> -1.5652 - 1.04343 I, y -> -0.319448 + 1.63317 I},

>     {x -> -1.5652 + 1.04343 I, y -> -0.319448 - 1.63317 I}}
```

Series

```
In[1]:= Sum[ Exp[-mu] mu^n / Factorial[n], {n, 0, Infinity} ]
Out[1]= 1

In[2]:= Sum[ n Exp[-mu] mu^n / Factorial[n], {n, 0, Infinity} ]
Out[2]= mu

In[3]:= Sum[ (n - mu)^2 Exp[-mu] mu^n / Factorial[n], {n, 0, Infinity} ]
Out[3]= mu
```

```
In[4]:= Sum[ p^k, {k, 0, n} ]

              1 + n
        -1 + p
Out[4]= -----------
          -1 + p
```

```
In[5]:= Sum[ p^k, {k, 1, n}]

                   n
        p (-1 + p )
Out[5]= -----------
          -1 + p
```


Limits

```
In[1]:= Limit[ Sin[x]/x, x -> 0 ]
Out[1]= 1

In[2]:= Limit[ 1/x, x -> Infinity ]
Out[2]= 0

In[3]:= Limit[ 1/x, x->0, Direction -> -1 ]
Out[3]= Infinity

In[4]:= Limit[ 1/x, x->0, Direction -> 1 ]
Out[4]= -Infinity
```

Integrals & derivatives

```
In[1]:= Integrate[ x^4 Cos[x], x ]
                   2                      2    4
Out[1]= 4 x (-6 + x ) Cos[x] + (24 - 12 x  + x ) Sin[x]
```

```
In[2]:= D[%, x]
          2                    2                      2    4
Out[2]= 8 x  Cos[x] + 4 (-6 + x ) Cos[x] + (24 - 12 x  + x ) Cos[x] -
               2                        3
>    4 x (-6 + x ) Sin[x] + (-24 x + 4 x ) Sin[x]
```

```
In[3]:= Simplify[%]
         4
Out[3]= x  Cos[x]
```

```
In[4]:= Integrate[ Exp[x], {x, -1, 1} ]
            1
Out[4]= -(-) + E
            E
```

```
In[5]:= Together[%]
                2
        -1 + E
Out[5]= -------
           E
```


Another example

Consider $X_1, X_2, X_3 \sim$ iid $N(\mu, \sigma^2)$.

Define $R = [X_{(2)} - X_{(1)}]/[X_{(3)} - X_{(1)}]$.

One can show that the density of R is $f(r) = \dfrac{3\sqrt{3}}{2\pi} \cdot \dfrac{1}{r^2 + r(1-r) + (1-r)^2}$

Find the cdf.

```
In[1]:= Integrate[ 3 Sqrt[3]/(2 Pi) / (r^2 + r(1-r) + (1-r)^2), r]

                  -1 + 2 r
        3 ArcTan[--------]
                  Sqrt[3]
Out[1]= ------------------
                Pi
```

```
In[2]:= % /. r -> 0
          1
Out[2]= -(-)
          2
```

```
In[3]:= Solve[%1 + 1/2 == 0.025, r]
Out[3]= {{r -> 0.0297866}}
```

Summary

- Mathematica can be useful for dealing with some tedious algebra or calculus.
- It is not really a substitute for thinking.
- Buy (or borrow) a book, or look for tutorials online.